# $Id: PKGBUILD 141061 2011-10-22 20:45:31Z eric $
# Maintainer: Jeff Mickey <jeff@archlinux.org>

pkgname=liquidwar
pkgver=5.6.4
pkgrel=6
pkgdesc="A unique multiplayer wargame"
arch=('i686' 'x86_64')
url="http://www.ufoot.org/liquidwar/"
license=('GPL')
depends=('allegro4')
makedepends=('python2')
install=liquidwar.install
source=(http://www.ufoot.org/download/$pkgname/v5/$pkgver/$pkgname-$pkgver.tar.gz
        http://www.ufoot.org/sig/$pkgname-$pkgver.tar.gz.sig liquidwar-5.6.4-ovflfix.patch)
sha1sums=('a3006c4da9517c3af3b7bf165b2c86fa369a71c6'
          'a262967d60e0f04ac5b99693144ef6b0038bb6c4'
          'e2102792c54a01b7cd14adb6d72404044175ec7e')

build() {
  cd "$srcdir/$pkgname-$pkgver"
  patch -p0 -i "$srcdir/liquidwar-5.6.4-ovflfix.patch"
  sed -i 's/python/python2/' doc/Makefile.in
  [ "$CARCH" == "x86_64" ] && EXTRAOPTS="--disable-asm"
  LIBS='-lm' PYTHON=yes ./configure --prefix=/usr $EXTRAOPTS
  make GAMEDIR=/usr/bin DATADIR=/usr/share/liquidwar
}

package() {
  cd "$srcdir/$pkgname-$pkgver"
  make DESTDIR="$pkgdir" GAMEDIR=/usr/bin DATADIR=/usr/share/liquidwar install_nolink
}
